
      ⍝ Train Spotting in V14
      ⍝ John Scholes - Dyalog

⍝     ⍺(f g h)⍵ → (⍺ f ⍵) g (⍺ h ⍵)           [fgh]
⍝     ⍺(A g h)⍵ →    A    g (⍺ h ⍵)           [Agh]
⍝     ⍺(  g h)⍵ →         g (⍺ h ⍵)            [gh]
                                                   
⍝      (f g h)⍵ → (  f ⍵) g (  h ⍵)           [fgh]
⍝      (A g h)⍵ →    A    g (  h ⍵)           [Agh]
⍝      (  g h)⍵ →         g (  h ⍵)            [gh]

      ⎕ ← col ← ⍪2 2  
2
2
      6(÷,-)col                             ⍝ [fgh]
3 4
3 4
      6(0,-)col                             ⍝ [Agh]
0 4
0 4
      6( ,-)col                             ⍝  [gh]
4 4
       (÷,-)col                             ⍝ [fgh]
0.5 ¯2
0.5 ¯2
       (0,-)col                             ⍝ [Agh]
0 ¯2
0 ¯2
       ( ,-)col                             ⍝  [gh]
¯2 ¯2

      ⍝ Trains associate right:
      ⍝   f g h j k → f g (h j k)
      ⍝     g h j k →   g (h j k)



      ]boxing on                                                   
Was ON
      +++++++
┌─┬─┬─────────────┐
│+│+│┌─┬─┬───────┐│
│ │ ││+│+│┌─┬─┬─┐││
│ │ ││ │ ││+│+│+│││
│ │ ││ │ │└─┴─┴─┘││
│ │ │└─┴─┴───────┘│
└─┴─┴─────────────┘
      ++++++
┌─┬─────────────┐
│+│┌─┬─┬───────┐│
│ ││+│+│┌─┬─┬─┐││
│ ││ │ ││+│+│+│││
│ ││ │ │└─┴─┴─┘││
│ │└─┴─┴───────┘│
└─┴─────────────┘

⍝ Identities:

⍝   A∘g B ←→ A g B      ⍝ left arg currying
⍝   A∘g h ←→ A g h      ⍝ atop ←→ fork

⍝  (f g)h ←→ f(g h)     ⍝ 2-trains are associative
⍝  (f g)h ←→ f∘g h      ⍝ atop ←→ composition

⍝ Binding issues:
⍝ V14 will retain classic Dyalog "early binding".
⍝ Name replaced by its value "on encounter".

      sum ← +⌿
      avg ← sum ÷ ≢   ⍝ uses _value_ of sum



      avg
┌─────┬─┬─┐
│┌─┬─┐│÷│≢│
││+│⌿││ │ │
│└─┴─┘│ │ │
└─────┴─┴─┘
      sum ← ⍟(×/*)    ⍝ changing sum
      avg             ⍝ does not change avg
┌─────┬─┬─┐
│┌─┬─┐│÷│≢│
││+│⌿││ │ │
│└─┴─┘│ │ │
└─────┴─┴─┘

⍝ Early binding avoids this name-capture problem:

      sum ← +
      tot ← sum/
      foo←{sum←⍺+⍵ ⋄ tot sum}  
      1 2 3 foo 4 5 6
21

⍝ Train examples:

      vec ← 3 1 4 1 5 9 2 6 5

      notin ← ~∊                ⍝ not-member
      vec notin 1 2 3
0 0 1 0 1 1 0 1 1
      minmax ← ⌊/ , ⌈/          ⍝ min-max
      minmax vec
1 9
      fstlst ← ⊃,(⊃⌽)           ⍝ first-last
      fstlst vec
3 5
      dice ← ?⍴                 ⍝ dice



      4 dice 10
6 1 7 5
      intdiv ← ⌊÷               ⍝ int division
      17 intdiv 5
3
      sort ← (⊂⍋)⌷⊢             ⍝ sort
      sort vec
1 1 2 3 4 5 5 6 9

⍝ Optimised:

      vec
3 1 4 1 5 9 2 6 5
      vec(⍳∘1 >)3               ⍝ (vec>3)⍳1
3
      vec(1 ⍳⍨ >)3              ⍝ (vec>3)⍳1
3
      vec(1 ∊ >)3               ⍝ 1∊vec>3
1
      vec(∨/ >)3                ⍝ ∨/vec>3
1

⍝ All of the above with: <≤=≥≠ for >
⍝   ..      ..     with:     0 for 1 
⍝   ..      ..     with: +/ ∧/ for ∨/

⍝ And finally:

      1 (⊣+÷)⍣= 1               ⍝ Phi
1.61803
      (0.5 × ⊢+÷)⍣=⍨ 2          ⍝ sqrt 2    ⍝ RH
1.41421

      j ← {⍺←0 ⋄ ⍺+0j1×⍵} ⋄ 3 j 4           ⍝ KEI
3J4
      cos←2∘○ ⋄ sin←1∘○



      Euler ← *∘j = cos j sin               ⍝ RH

      ⎕ ← M ← 0.01 × ¯100 + 4 5 dice 200
 0.41  0.83 0.53 ¯0.47 ¯0.9 
 0.48 ¯0.34 0.27  0.52  0.99
¯0.26 ¯0.5  0.97  0.45  0.51
 0.31 ¯0.85 0.27  0.77 ¯0.45

      Euler M
1 1 1 1 1
1 1 1 1 1
1 1 1 1 1
1 1 1 1 1
      Euler
┌─────────────────────┬─┬─────────────────────────────────┐
│┌─┬─┬───────────────┐│=│┌───────┬───────────────┬───────┐│
││*│∘│{⍺←0 ⋄ ⍺+0J1×⍵}││ ││┌─┬─┬─┐│{⍺←0 ⋄ ⍺+0J1×⍵}│┌─┬─┬─┐││
│└─┴─┴───────────────┘│ │││2│∘│○││               ││1│∘│○│││
│                     │ ││└─┴─┴─┘│               │└─┴─┴─┘││
│                     │ │└───────┴───────────────┴───────┘│
└─────────────────────┴─┴─────────────────────────────────┘



⍝ Compiling single-line dfns into trains:

       parse'1+2*3×4÷...'
┌─+─┐       
1 ┌─*─┐     
  2 ┌─×─┐   
    3 ┌─÷─┐ 
      4  ...

⍝   {L f R ...} → ({L} f {R ...})               [D]
⍝   {  f R ...} → (    f {R ...})               [M]
⍝ with base cases:
⍝           {⍺} → ⊣                             [⍺]
⍝           {⍵} → ⊢                             [⍵]
⍝ and optimisations:
⍝     ({A}g{C}) → {A g C}  ⍝ evaluating A g C   [⍎]
⍝      ({A}g h) → (A g h)                       [A]
⍝      (f g{C}) → (C g⍨ f)                      [C]      
⍝      (f(g h)) → (f∘g h)                       [∘]
⍝          (f⊢) → f        ⍝ for monadic f⊢     [⊢]

      {(+⌿⍵)  ÷ ≢⍵}   3 4       ⍝ dfn     
3.5
      ({+⌿⍵}  ÷{≢⍵})  3 4       ⍝ [D] 
3.5
      ((+⌿{⍵})÷(≢{⍵}))3 4       ⍝ [M] [M]
3.5
      ((+⌿ ⊢ )÷(≢ ⊢ ))3 4       ⍝ [⍵] [⍵]
3.5
      ( +⌿    ÷ ≢    )3 4       ⍝ [⊢] [⊢]
3.5

⍝ That's All Folks ...


